CALCULATION OF THE DISTRIBUTION FUNCTION OF INFINITE QUADRATIC FORMS IN NORMAL VARIABLES

by

Alan J. Lee

University of Auckland
and
University of North Carolina at Chapel Hill

Abstract

Simple but effective methods of computing the distribution function random variables of the form $\sum_{n=1}^{\infty} \lambda_n Z_n^2$ (Z_n i.i.d. $N(0,1)$) are described. A FORTRAN subroutine to implement the methods is appended. Some examples are given.

Key Words and Phrases: infinite quadratic form in normal variables, numerical calculation of distribution functions

---

The work of this author was supported by the Air Force Office of Scientific Research under Grant AFOSR-75-2796.

Introduction.

In many places in statistics (for example, Durbin and Knott (1972); Blum, Kiefer and Rosenblatt (1961); Hoeffding (1948)) one must compute the distribution function F of the infinite quadratic form $X = \sum_{n=1}^{\infty} \lambda_n Z_n^2$, where the coefficients satisfy $\sum_{n=1}^{\infty} \lambda_n < \infty$ and the Z_n are independent N(0,1) random variables.

In certain applications of the theory of U-statistics (see e.g. Lee (1979)), the coefficients λ_n are not available in closed form but are known to be the eigenvalues of a certain integral operator. In this case the λ_n must be determined numerically. In other applications, the λ_n are explicitly known and the characteristic function ϕ corresponding to F, namely $\phi(t) = \prod_{n=1}^{\infty} (1-2i\lambda_n t)^{-\frac{1}{2}}$, may be expressed in closed form. F may then be obtained by numerical inversion of the c.f. ϕ. (For techniques of numerical inversion, see e.g. Bohman (1972), Davies (1973).) If the infinite product defining ϕ cannot be written in closed form, the technique to be described may still be used to advantage.

Accordingly, it seems useful to have a method for the evaluation of F that depends on knowledge of the first N eigenvalues (arranged in descending order of magnitude) and also on the four quantities (traces)

$$S_j = \sum_{n=1}^{\infty} \lambda_n^j ,$$

which may be obtained from the kernel of the integral operator without explicit determination of the λ_n (in the case that all the λ_n are not known) or by simple calculation in the case where the λ_n are known explicitly.

Description of the methods.

A naive approach is to approximate F by $F^{(N)}$, where F_N is the d.f. corresponding to the truncated c.f.

$$\phi^{(N)}(t) = \prod_{n=1}^{N} (1-2i\lambda_n t)^{-\frac{1}{2}}$$

$F^{(N)}$ is recovered from $\phi^{(N)}$ by numerical inversion. However, unless the quadratic form is finite, and all eigenvalues are used, this method may lead to substantial inaccuracy. (For a discussion, see Blum, Kiefer and Rosenblatt (1961).) A much better method, advocated by Durbin and Knott (1972), is to approximate F by F_1, where F_1 is the d.f. of a random variable of the form

$$X_1 = \sum_{n=1}^{N} \lambda_n Z_n^2 + cY$$

where Y is an χ^2 variate with q degrees of freedom, and c and q are chosen to make the first two cumulants of X and X_1 agree, i.e.

$$c = S_{2,N}/S_{1,N} ,$$

$$q = S_{1,N}^2/S_{2,N} ,$$

where $S_{j,N} = S_j - \sum_{n=1}^{N} \lambda_n^j$. One then computes the c.f. of X_1, namely $\phi_1(t) = (1-2ict)^{-q/2} \prod_{n=1}^{N} (1-2i\lambda_n t)^{-\frac{1}{2}}$, and recovers the approximating F_1 by numerical inversion of ϕ_1.

This method is effective when all λ_n are positive, but one expects inaccuracies if substantial numbers of the λ_n, $n > N$, are negative. A modification of the method consists of using the approximation F_2 where F_2 is the d.f. of the r.v.

$$X_2 = \sum_{n=1}^{N} \lambda_n Z_n^2 + \alpha Y_1 + \beta Y_2$$

where Y_1, Y_2 are χ^2 variates with q_1 and q_2 d.f., independent of each other and the Z_n. To make the first four cumulants of X and X_2 coincide, the parameters are chosen to be the solutions of the equations

$$\alpha^j q_1 + \beta^j q_2 = S_{j,N} \qquad j = 1,\ldots,4 .$$

Elementary algebra shows that α and β are the roots of the quadratic

$$a^2(S_{1,N}S_{3,N} - S_{2,N}^2) + a(S_{3,N}S_{2,N} - S_{1,N}S_{4,N}) + (S_{4,N}S_{2,N} - S_{3,N}^2) = 0 .$$

Assuming α,β to be real, we then have

$$q_1 = (S_{2,N} - \beta S_{1,N})/\alpha(\alpha - \beta)$$

$$q_2 = (\alpha S_{1,N} - S_{2,N})/\beta(\alpha - \beta) .$$

Assuming q_1, q_2 to be positive, we approximate ϕ by $\phi_2(t) = (1-2i\alpha t)^{-q_1/2}(1-2i\beta t)^{-q_2/2}\prod_{i=1}^{N}(1-2i\lambda_n t)^{-\frac{1}{2}}$ and invert ϕ_2 to obtain F_2.

The numerical inversion technique used in the program in the appendix is that of Davies (1973), which approximates the d.f. F_2 at x by

$$\tfrac{1}{2} - \frac{\Delta}{\pi} \sum_{k=0}^{K} \mathrm{Im}[\phi_2((k+\tfrac{1}{2})\Delta)\exp -(i(k+\tfrac{1}{2})\Delta x)]/(k+\tfrac{1}{2})\Delta \ .$$

If ε is the maximum permitted error in the inversion, the spacing parameter Δ and the truncation parameter K are chosen to make

$$\max(\Pr[X_2 > x + 2\pi/\Delta] \ , \Pr[X_2 < x - 2\pi/\Delta]) < \varepsilon/2$$

and the truncation error in the series is less than $\varepsilon/2$. Suitable values of K are easily chosen by noting the sensitivity of the results to different values. The choice of Δ is facilitated by the estimates

$$\max(\Pr[X_2 > \delta_1] \ , \Pr[X_2 < -\delta_2]) \le \Pr[|X_2| > \delta]$$

where $\delta_1, \delta_2 > 0$ and $\delta = \min(\delta_1, \delta_2)$. Now

$$\begin{aligned} \Pr[|X_2| > \delta] &\le \Pr[\sum_{n=1}^{N} |\lambda_n| Z_n + |\alpha| Y_1 + |\beta| Y_2 > \delta] \\ &\le \exp(-\frac{S_1^*}{2\lambda^*}(\frac{\delta}{S_1^*} - 1 - \log(\delta/S_1^*)) \end{aligned} \qquad (1)$$

where the last estimate is obtained using the methods of Blum <u>et al</u>. (1961) and

$$S_1^* = \sum_{n=1}^{N} |\lambda_n| + |\alpha| + |\beta|$$

$$\lambda^* = \max(|\lambda_n| \ , n=1,\ldots,N \ , |\alpha| \ , |\beta|) \ .$$

Thus we choose Δ so that $x + 2\pi/\Delta$ and $x - 2\pi/\Delta$ are respectively positive and negative, and at least δ in magnitude, where δ is such that (1) is less than $\varepsilon/2$ in magnitude.

Examples.

The three techniques (simple truncation, single χ^2 approximation, double χ^2 approximation) were tried on two examples with the results noted below.

Example 1. Let the coefficients λ_n be given by $\lambda_{2n} = 1/\pi^2 n^2$, $\lambda_{2n-1} = 1/\pi^2 n^2$, $n = 1,2,\ldots$. The characteristic function for $t > 0$ is given by $\phi(t) = \prod_{n=1}^{\infty} (1 - 2it/\pi^2 n^2)^{-1} = \frac{z}{\sin z}$ where $z = (1+i)\sqrt{t}$. The distribution function is $F(x) = 1 + 2 \sum_{n=1}^{\infty} (-1)^n \exp(-\pi^2 n^2 x/2)$. Table I below gives the results of using the program in the appendix.

TABLE I

Results rounded to 5 decimal places, $\Delta = 2$, $K = 200$.
Calculation of $F(x)$ for selected x-values.

	x = .2	x = .5	x = 1.0	x = 1.5
Exact	.29290	.83049	.98562	.99878
Numerical Inversion	.29290	.83049	.98562	.99878
Truncation				
N = 4	.50949	.88694	.99041	.99919
N = 10	.39715	.85871	.98801	.99898
N = 20	.34855	.84588	.98692	.99889
Single χ^2				
N = 4	.29228	.83054	.98562	.99878
N = 6	.29280	.83050	.98562	.99878
N = 8	.29287	.83050	.98562	.99878
Double χ^2				
N = 4	.29290	.83049	.98562	.99878
N = 6	.29290	.83049	.98562	.99878
N = 8	.29290	.83049	.98562	.99878

Example 2. Let now the coefficients be given by $\lambda_n = (-1)^{n-1}/\pi^2 n^2$. The characteristic function is for $t > 0$

$$\phi(t)=[\cos z \ \overline{\sin hz} / \overline{z}\,]^{-\frac{1}{2}}, \ z=(1+i)\sqrt{t}/2 \ .$$

Table II below gives the results of computations using the program in the appendix.

TABLE II

Results rounded to 5 decimal places, $\Delta = 2$, $K = 200$.
Calculation of $F(x)$ for selected x-values.

	x = 0.0	x = .5	x = 1.0	x = 1.5
Numerical Inversion	.25506	.97564	.99844	.99989
Truncation				
N = 4	.27013	.97599	.99846	.99989
N = 10	.25820	.97570	.99844	.99989
N = 20	.25592	.97565	.99844	.99989
Single χ^2				
N = 4	.25602	.97563	.99844	.99989
N = 6	.25557	.97563	.99844	.99989
N = 8	.25533	.97564	.99844	.99989
Double χ^2				
N = 4	.25507	.97564	.99844	.99989
N = 6	.25506	.97564	.99844	.99989
N = 8	.25506	.97564	.99844	.99989

Remarks. Truncation is obviously unsatisfactory, except possibly on the extreme tails of the distributions. The single χ^2 approximation is a considerable improvement, especially when all coefficients are positive.

However, the double χ^2 approximation is clearly superior except in the tails, giving results correct to 5 decimals with only four eigenvalues in the first example and within one digit in the fifth decimal place in the second.

APPENDIX

The FORTRAN program below calculates either the truncation, single χ^2 or double χ^2 approximation to F. The subroutine employs the following formal parameters:

<u>Input</u>

EIG:	vector of N coefficients λ_n
N:	dimension of EIG
CUM:	4-vector of quantities S_j
X:	m-vector of arguments x_i of distribution function
M:	dimension of X
IND:	set to 0 for truncation, 1 for single χ^2 and 2 for double χ^2 approximation
DELTA:	spacing parameter Δ
KTRUNC:	truncation parameter K

<u>Output</u>

F:	m-vector of m function values $F(x_i)$
IER:	error indicator. Set to 0 if double χ^2 approximation fails.

The program should give satisfactory results in single precision on a 32-bit machine.

```
      SUBROUTINE DISTRB(EIG,N,CUM,X,M,F,IND,DELTA,KTRUNC,IER)
      DIMENSION EIG(N),X(M),CUM(4),SUM(4),S(4)
      REAL IMAG,IMAG1
      DATA PI/3.141592654/
      IER=1
      IF(IND.EQ.0)GO TO 7
C
C     NOW SET UP FOR CHI-SQUARE APPROXIMATIONS
C
      LIM=2*IND
      DO 1 I=1,LIM
1     S(I)=CUM(I)
      DO 3 N1=1,N
      TM=1.0
      DO 2 J=1,4
      TM=TM*EIG(N1)
2     S(J)=S(J)-TM
3     CONTINUE
      IF(IND.EQ.1)GO TO 55
C
C     NOW SET UP FOR DOUBLE CHI-SQUARE APPROXIMATIONS
C
      Z1=S(1)/S(2)
      Z3=S(3)/S(2)
      Z4=S(4)/S(2)
      A=Z3*Z1-1.0
      B=Z3-Z1*Z4
      C=Z4-Z3*Z3
      D=B*B-4.0*A*C
      IF(D.GE.0.0)GO TO 5
C
C     EXIT IF DOUBLE APPROXIMATION FAILS.
C
      IER=0
      RETURN
C
C     CALCULATE PARAMETERS OF DOUBLE APPROXIMATION
C
5     AL=-(B+SQRT(D))/(2.0*A)
      BE=(Z3-AL)/(1.0-AL*Z1)
      XX=(1.0-BE*Z1)/(AL-BE)
      Q1=S(2)*XX/AL
      Q2=(Z1-XX)*S(2)/BE
      IF(Q1.LE.0.0)GO TO 50
      IF(Q2.GT.0.0)GO TO 7
C
C     EXIT IF DOUBLE APPROXIMATION FAILS.
C
50    IER=0
      RETURN
C
C     CALCULATE PARAMETERS OF SINGLE APPROXIMATION
C
```

```
55    C=S(2)/S(1)
      Q=S(1)*S(1)/S(2)
C
C     BEGIN NUMERICAL INVERSION
C
7     DO 9 I=1,M
9     SUM(I)=0.0
      DO 8 K=0,KTRUNC
      T=(K+.5)*DELTA
C
C     COMPUTE C.F. OF FINITE QUADRATIC FORM
C
      REAL=1.0
      IMAG=0.0
      DO 20 L=1,N
      S3=2.0*T*EIG(L)
      T3=ATAN(-S3)
      RHO3=SQRT(SQRT(1.0+S3*S3))
      REAL1=COS(T3/2.0)/RHO3
      IMAG1=-SIN(T3/2.0)/RHO3
      REAL2=REAL*REAL1-IMAG*IMAG1
      IMAG=IMAG*REAL1+REAL*IMAG1
      REAL=REAL2
20    CONTINUE
      IF(IND.EQ.0)GO TO 10
      IF(IND.EQ.2)GO TO 11
C
C     ADJUST C.F. FOR SINGLE APPROXIMATION
C
      CALL CHISQ(T,REAL,IMAG,C,Q)
      GO TO 10
C
C     ADJUST C.F. FOR DOUBLE APPROXIMATION
C
11    CALL CHISQ(T,REAL,IMAG,AL,Q1)
      CALL CHISQ(T,REAL,IMAG,BE,Q2)
C
C     COMPLETE NUMERICAL INVERSION
C
10    DO 12 I=1,M
      SUM(I)=SUM(I)+(IMAG*COS(T*X(I))-REAL*SIN(T*X(I)))/T
12    CONTINUE
8     CONTINUE
      DO 14 I=1,M
      F(I)=.5-SUM(I)*DELTA/PI
14    CONTINUE
      RETURN
      END
      SUBROUTINE CHISQ(T,REAL,IMAG,C,Q)
C
C     SUBROUTINE MULTIPLIES THE COMPLEX NUMBER (REAL,IMAG) BY (1-C*I*T)*(-Q/2.0)
      AND RETURNS RESULT AS (REAL,IMAG).
C
```

```
REAL IMAG,IMAG1
S=-2*C*T
R=SQRT(1.0+S*S)
THETA=ATAN(S)*(-Q/2.0)
RHO=R**(-Q/2.0)
REAL1=RHO*COS(THETA)
IMAG1=RHO*SIN(THETA)
REAL2=REAL*REAL1-IMAG*IMAG1
IMAG=REAL1*IMAG+REAL*IMAG1
REAL=REAL2
RETURN
END
```

References

Blum, J.R., Kiefer, J., Rosenblatt, M. (1961). Distribution-free tests of independence based on the sample distribution function. *Ann. Math. Statist. 32*, 485-498.

Bohmann, H. (1972). From the characteristic function to the distribution function via Fourier Analysis. *BIT 12*, 279-283.

Davies, R.B. (1973). Numerical inversion of the characteristic function. *Biometrika 60*, 415-417.

Durbin, J. and Knott, M. (1972). Components of Cramér - Von Mises statistics I. *J.R. Statist. Soc. 34*, 290-307.

Hoeffding, W. (1948). A class of statistics with asymptotically normal distribution. *Ann. Math. Statist. 19*, 293-325.

Lee, A.J. (1979). On the asymptotic distribution of U-statistics. *Institute of Statistics Mimeo Series #1255*, University of North Carolina at Chapel Hill.

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER	2. GOVT ACCESSION NO. / 3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) Calculation of the Distribution Function of Infinite Quadratic Forms in Normal Variables	5. TYPE OF REPORT & PERIOD COVERED TECHNICAL
	6. PERFORMING ORG. REPORT NUMBER Mimeo Series No. 1274
7. AUTHOR(s) Alan J. Lee	8. CONTRACT OR GRANT NUMBER(s) Contract AFOSR-75-2796
9. PERFORMING ORGANIZATION NAME AND ADDRESS	10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS
11. CONTROLLING OFFICE NAME AND ADDRESS Air Force Office of Scientific Research Bolling Air Force Base Washington, DC 20332	12. REPORT DATE March 1980
	13. NUMBER OF PAGES 12
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office)	15. SECURITY CLASS. (of this report) UNCLASSIFIED
	15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for Public Release -- Distribution Unlimited

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

infinite quadratic form in normal variables
numerical calculation of distribution functions

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

Simple but effective methods of computing the distribution function random variables of the form $\sum_{n=1}^{\infty} \lambda_n Z_n^2$ (Z_n i.i.d. $N(0,1)$) are described. A FORTRAN subroutine to implement the methods is appended. Some examples are given.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

SECURITY CLASSIFICATION OF THIS PAGE(*When Data Entered*)

SECURITY CLASSIFICATION OF THIS PAGE(*When Data Entered*)